
Overview of Topic VI

This topic will cover

• One-Way Analysis of Variance (ANOVA)

– Single-Factor Study (§16)

– Analysis of Factor Level Means (§17)

– Diagnostics and Remedial Measures (§18)

One-Way Analysis of Variance (ANOVA)

• Also called “single factor ANOVA”.

• The response variable Y is continuous (same as in regression).

• There are two key differences regarding X .

1. It is a qualitative variable (e.g. gender, location, etc). Instead

of calling it an explanatory variable, we now refer to it as a

factor .

2. No assumption (i.e. linear relationship) is made about the

nature of the relationship between X and Y . Rather we

attempt to determine whether the response differ significantly

at different levels of X . This is a generalization of the

two-independent-sample t-test .

Topic VI

Page 2



Statistics 512: Applied Regression Analysis

Professor Min Zhang

Purdue University

Spring 2014

• We will have several different ways of parameterizing the model:

1. the cell means model

2. the factor effects model

– two different possible constraint systems for the factor

effects model

Notation for One-Way ANOVA

X (or A) is the qualitative factor

• r (or a) is the number of levels

• we often refer to these as groups or treatments

Y is the continuous response variable

• Yi,j is the jth observation in the ith group.

• i = 1, 2, . . . , r levels of the factor X .

• j = 1, 2, . . . , ni observations at factor level i.

KNNL Example (page 685)

• See the file nknw677.sas for the SAS code.

• Y is the number of cases of cereal sold (CASES )

• X is the design of the cereal package (PKGDES )

• There are 4 levels for X representing 4 different package

designs: i = 1 to 4 levels

• Cereal is sold in 19 stores, one design per store. (There were

originally 20 stores but one had a fire.)

• j = 1, 2, . . . , ni stores using design i. Here ni = 5, 5, 4, 5.

We simply use n if all of the ni are the same. The total number

of observations is nT =
∑r

i=1
ni = 19.

data cereal;

infile ’H:\System\Desktop\CH16TA01.DAT’;

input cases pkgdes store;

proc print data=cereal;

Obs cases pkgdes store

1 11 1 1

2 17 1 2

3 16 1 3

4 14 1 4

5 15 1 5

6 12 2 1

7 10 2 2

8 15 2 3

9 19 2 4

10 11 2 5

11 23 3 1

12 20 3 2

13 18 3 3

14 17 3 4

15 27 4 1

16 33 4 2

17 22 4 3

18 26 4 4

19 28 4 5

Note that the “store ” variable is just j; here it does not label a particular store,

and we do not use it (only one design per store).

Model (Cell Means Model)

Model Assumptions

• Response variable is normally distributed

• Mean may depend on the level of the factor

• Variance is constant

• All observations are independent

Cell Means Model

Yi,j = µi + ǫi,j

• µi is the theoretical mean of all observations at level

i.

• ǫi,j ∼iid N(0, σ2) and hence Yi,j ∼iid N(µi, σ
2).

• Note there is no “intercept” term and we just have a

potentially different mean for each level of X . In this

model, the mean does not depend numerically on the

actual value of X (unlike the linear regression model).

Parameters

• The parameters of the model are µ1, µ2, . . . , µr, σ
2.

• Basic analysis question is whether or not the explanatory

variable helps to explain the mean of Y . In this case, this is the

same as asking whether or not µi depends on i. So we will want

to test H0 : µ1 = µ2 = . . . = µr against the alternative

hypothesis that the means are not all the same.

We may further be interested in grouping the means into

subgroups that are equivalent (statistically indistinguishable).

Estimates

• Estimate µi by the mean of the observations at level i. That is,

µ̂i = Ȳi. =

∑

j Yi,j

ni

• For each level i, get an estimate of the variance,

s2i =

∑ni

j=1
(Yi,j − Ȳi.)

2

ni − 1

• We combine these s2i to get an estimate of σ2 in the following

way.

Pooled Estimate of σ2

If the ni are all the same we would simply average the s2i ; otherwise use a

weighted average. (Do not average the si.) In general we pool the s2i , using

weights proportional to the degrees of freedom ni − 1 for each group. So the

pooled estimate is

s2 =

∑r

i=1(ni − 1)s2i
∑r

i=1(ni − 1)
=

∑r

i=1(ni − 1)s2i
nT − r

=

∑r

i=1

∑ni

j=1(Yi,j − Ȳi.)
2

nT − r

= MSE.

In the special case that there are an equal number of observations per group

(ni = n) then nT = nr and this becomes s2 =
(n−1)

∑

r

i=1
s2
i

nr−r
= 1

r

∑r

i=1 s
2
i ,

a simple average of the s2i .

Run proc glm

glm standards for “General Linear Model”. The

class statement tells proc glm that pkgdes is a

“classification” variable, i.e. categorical. The

class statement defines variables which are

qualitative in nature. The means statement requests

sample means and standard deviations for each factor

level.

proc glm data=cereal;

class pkgdes;

model cases=pkgdes;

means pkgdes;

The GLM Procedure

Class Level Information

Class Levels Values

pkgdes 4 1 2 3 4

Number of observations 19

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 3 588.2210526 196.0736842 18.59 <.0001

Error 15 158.2000000 10.5466667

Corrected Total 18 746.4210526

R-Square Coeff Var Root MSE cases Mean

0.788055 17.43042 3.247563 18.63158

means statement output

Level of ------------cases------------

pkgdes N Mean Std Dev

1 5 14.6000000 2.30217289

2 5 13.4000000 3.64691651

3 4 19.5000000 2.64575131

4 5 27.2000000 3.96232255

Plot the data.

symbol1 v=circle i=none;

proc gplot data=cereal;

plot cases*pkgdes;

Look at the means and plot them.

proc means data=cereal;

var cases; by pkgdes;

output out=cerealmeans mean=avcases;

proc print data=cerealmeans;

Obs pkgdes _TYPE_ _FREQ_ avcases

1 1 0 5 14.6

2 2 0 5 13.4

3 3 0 4 19.5

4 4 0 5 27.2

Topic VI

Page 16



Statistics 512: Applied Regression Analysis

Professor Min Zhang

Purdue University

Spring 2014

symbol1 v=circle i=join;

proc gplot data=cerealmeans;

plot avcases*pkgdes;

Some more notation

• The mean for group or treatment i is Ȳi. =
∑ni

j=1
Yi,j

ni
.

• The overall of “grand” mean is Ȳ.. =
∑r

i=1

∑ni
j=1

Yi,j

nT
.

• The total number of observations is nT =
∑r

i=1
ni.

ANOVA Table

Source df SS MS

Reg r − 1
∑

i(Ȳi. − Ȳ..)
2 SSR

dfR

Error nT − r
∑

i,j(Yi,j − Ȳi.)
2 SSE

dfE

Total nT − 1
∑

i,j(Yi,j − Ȳ..)
2 SST

dfT

Expected Mean Squares

• E(MSR) = σ2 +
∑

i ni(µi−µ.)
2

r−1 , where µ. =
∑

i niµi

nT
.

• E(MSE) = σ2.

• E(MSR) > E(MSE) when some group means

are different.

• See KNNL pages 694 - 696 for more details. In more

complicated models, these tell us how to construct

the F -test.

F -test

H0 : µ1 = µ2 = . . . = µr

Ha : not all µi are equal

F =
MSR

MSE

• Under H0, F ∼ F(r−1,nT−r)

• Reject H0 when F is large.

• Report the p-value.

Factor Effects Model

The factor effects model is just a re-parameterization of

the cell means model. It is a useful way at looking at

more complicated models; for now it may not seem

worth the trouble but it will be handy later. Often the null

hypotheses are easier to interpret with the factor effects

model. The model is Yi,j = µ+ τi + ǫi,j where

ǫi,j ∼iid N(0, σ2).

Parts of the Model

• µ is the overall or grand mean (it looks like an

intercept). Note: The text calls this µ., a notation I will

not use in the notes.

• The τi represent the difference between the overall

mean and the mean for level i. So whereas the cell

means model looks at the mean for each level, this

model looks at the amount by which the mean at each

level deviates from some “standard”.

Parameters

• The parameters of the factor effects model are µ, τ1, τ2, . . . , τr, σ
2. There

are r + 2 of these.

• Recall that the cell means model had r+1 parameters: µ1, µ2, . . . , µr, σ
2,

so in our new model one of the τ ’s is redundant. Thus we will need to place a

restraint on the τ ’s to avoid estimating this “extra” parameter. (The models

should be equivalent.)

• The relationship between the models is that µi = µ+ τi for every i. If we

consider the sum of these, we have
∑

µi = rµ+
∑

τi. If the ni are equal

this is just rµ = rµ+
∑

τi so the constraint we place on the model is
∑

τi = 0. Thus we need only estimate all of the τ ’s, except for one which

may be obtained from the others.

Constraints – An Example

Suppose r = 3, µ1 = 10, µ2 = 20, µ3 = 30. Without the

restrictions, we could come up with several equivalent sets of

parameters for the factor effects model. Some include

µ = 0, τ1 = 10, τ2 = 20, τ3 = 30 (same)

µ = 20, τ1 = −10, τ2 = 0, τ3 = 10

µ = 30, τ2 = −20, τ2 = −10, τ3 = 0

µ = 5000, τ1 = −4990, τ2 = −4980, τ3 = −4970

In this situation, these parameters are called not estimable or not well defined.

That is to say that there are many solutions to the least squares problem (not a

unique choice) and in fact the X′
X matrix for this parameterization does not

have an inverse. While there are many different restrictions that could be used

(e.g. µ = 0 would lead to the cell means model), the common restriction that
∑

i τi = 0 sets things up so that µ is the grand average and the τ ’s represent

the deviations from that average. This effectively reduces the number of

parameters by 1. The details are a bit more complicated when the ni are not all

equal; in that case it is appropriate to weight the terms in the sum by their relative

sample sizes. See KNNL pages 701-704 for details.

In summary, we always have µi = µ+ τi as the relationship between the cell

means model and the factor effects model. The constraint
∑

i τi = 0 implies

µ. =
1
r

∑

i µi (mean of group means). (The grand mean is the weighted mean

of group means, i.e., µ =
∑r

i=1 niµi/nT which implies the weighted zero-sum

constraint
∑r

i=1 niτi = 0.)

Hypothesis Tests

• The group or factor level effects are τi = µi − µ..

• The cell means model hypotheses were

H0 : µ1 = µ2 = . . . = µr

Ha : not all of the µi are equal

• For the factor effects model these translate to (when the

constrain is either τ1 + · · · + τr = 0 or τr = 0)

H0 : τ1 = τ2 = . . . = τr = 0

Ha : at least one of the τi is not 0

Estimators of Parameters

With the zero-sum constraint
∑

i τi = 0, the estimates are µ̂ = Ȳ..

and τ̂ = Ȳi. − Ȳ...

Solution used by SAS

Recall, X′
X may not have an inverse. We can use a generalized

inverse in its place. (X′
X)− is the standard notation for a

generalized inverse.

Definition: the generalized inverse of a matrix A is any matrix A−

satisfying AA
−
A = A. The generalized inverse is not unique.

There are many generalized inverses, each corresponding to a

different constraint (underdetermined system). The matrix A does

not have to be square.

The particular (X′
X)− used in proc glm corresponds to the

constraint τr = 0 (note this is different from our constraint). Recall

that µ and the τi are not uniquely estimable separately. But the

linear combinations µ+ τi are estimable. These are estimated by

the cell means model.

KNNL Example page 685

• The code is in the file nknw677.sas .

• Y is the number of cases of cereal sold

• X is the design of the cereal package

• i = 1 to 4 levels

• j = 1 to ni stores with design i

SAS Coding for X

SAS does this automatically/internally. You don’t need to do the

work to specify this in SAS.

The nT rows of the design matrix are copies of the following four

possible rows:

1 1 0 0 0 for level 1 (i.e. this is row i if Xi = 1)

1 0 1 0 0 for level 2

1 0 0 1 0 for level 3

1 0 0 0 1 for level 4

So our design matrix is

X =



































































































1 1 0 0 0

1 1 0 0 0

1 1 0 0 0

1 1 0 0 0

1 1 0 0 0

1 0 1 0 0

1 0 1 0 0

1 0 1 0 0

1 0 1 0 0

1 0 1 0 0

1 0 0 1 0

1 0 0 1 0

1 0 0 1 0

1 0 0 1 0

1 0 0 0 1

1 0 0 0 1

1 0 0 0 1

1 0 0 0 1

1 0 0 0 1



































































































The columns correspond to the parameter vector

β =























µ

τ1

τ2

τ3

τ4























.

You can see that the parameter µ acts a little like the

intercept parameter in regression.

Some options in proc glm

proc glm data=cereal;

class pkgdes;

model cases=pkgdes/xpx inverse solution;

Result of xpx option: the xpx option actually gives the

following matrix:





X ′X X ′Y

Y ′X Y ′Y





The X’X Matrix

Intercept pkgdes1 pkgdes2 pkgdes3 pkgdes4 cases

Intercept 19 5 5 4 5 354

pkgdes 1 5 5 0 0 0 73

pkgdes 2 5 0 5 0 0 67

pkgdes 3 4 0 0 4 0 78

pkgdes 4 5 0 0 0 5 136

cases 354 73 67 78 136 7342

Result of inverse option: the inverse option
actually gives the following matrix:
[

(X′
X)− (X′

X)−X′
Y

Y
′
X(X′

X)− Y
′
Y −Y

′
X(X′

X)−X′
Y

]

.

X’X Generalized Inverse (g2)

Intercept pkgdes1 pkgdes2 pkgdes3 pkgdes4 cases

Intercept 0.2 -0.2 -0.2 -0.2 0 27.2

pkgdes 1 -0.2 0.4 0.2 0.2 0 -12.6

pkgdes 2 -0.2 0.2 0.4 0.2 0 -13.8

pkgdes 3 -0.2 0.2 0.2 0.45 0 -7.7

pkgdes 4 0 0 0 0 0 0

cases 27.2 -12.6 -13.8 -7.7 0 158.2

Parameter estimates are in upper right corner; SSE is

lower right corner.

Parameter estimates (from solution option)

Standard

Parameter Estimate Error t Value Pr > |t|

Intercept 27.2 B 1.45235441 18.73 <.0001

pkgdes 1 -12.6 B 2.05393930 -6.13 <.0001

pkgdes 2 -13.8 B 2.05393930 -6.72 <.0001

pkgdes 3 -7.7 B 2.17853162 -3.53 0.0030

pkgdes 4 0.0 B . . .

Note that these are just the same estimates as in the

inverse matrix.

Caution Message

NOTE: The X’X matrix has been found to be

singular, and a generalized inverse

was used to solve the normal equations.

Terms whose estimates are followed by

the letter ’B’ are not uniquely

estimable.

Interpretation

If τr = 0 (in our case, τ4 = 0), then the corresponding

estimate should be zero. This means that the “intercept”

µ in SAS is estimated by the mean of the observations

in group 4. Since µ+ τi is the mean of group i, the τi

are the differences between the mean of group i and the

mean of group 4.

Level of ------------cases------------

pkgdes N Mean Std Dev

1 5 14.6000000 2.30217289

2 5 13.4000000 3.64691651

3 4 19.5000000 2.64575131

4 5 27.2000000 3.96232255

Parameter Estimates from means

µ̂ = 27.2 = 27.2

1 14.6 τ̂1 = 14.6− 27.2 = −12.6

2 13.4 τ̂2 = 13.4− 27.2 = −13.8

3 19.5 τ̂3 = 19.5− 27.2 = −7.7

4 27.2 τ̂4 = 27.2− 27.2 = 0

The means output gives the estimates of the µi (cell

means model). By subtracting off the mean for the last

level from each of these means we get estimates for the

factor effects (τ ’s) which match the results of the

solution option.

Bottom line: you can use SAS to automatically get

estimates for either the cell means model or the factor

effects model with the last τ = 0, i.e., τr = 0. You can

also use appropriate subtractions to get estimates for

any other constraint you choose. For example, if we

want to use

µ̂ = 5×14.6+5×13.4+4×19.5+5×27.2
19 = 354

19 = 18.63 then

subtract 18.63 from each of the µi estimates to get the

τi estimates.

Summary: Single Factor Analysis of Variance

Cell Means Model

Yi,j = µi + ǫi,j

• µi is the theoretical mean of all observations at level i

• ǫi,j ∼iid N(0, σ2) and hence Yi,j ∼iid N(µi, σ
2)

• Note there is no “intercept” term and we just have a potentially

different mean for each level of X . In this model, the mean does

not depend numerically on the actual value of X (unlike the

linear regression model).

With the cell means model there are no problems with parameter

estimability and matrix inversion. Use the means statement in

proc glm to get these estimates.

Factor Effects Model

Yi,j = µ+ τi + ǫi,j where ǫi,j ∼iid N(0, σ2)

• This is a reparameterization of the cell means model and a

useful way at looking at more complicated models.

• It is more useful since the null hypotheses are easier to

state/interpret. But there are problems with singularity of X′
X.

• We utilize a constraint (e.g.
∑

τi = 0 or in SAS τr = 0) to deal

with these problems.

Section 16.8: Regression Approach to ANOVA

Essentially one-way ANOVA is linear regression with

indicator (dummy) explanatory variables. We can use

multiple regression to reproduce the results based on

the factor effects model Yi,j = µ+ τi + ǫi,j where we

will restrict
∑

i τi = 0 by forcing τr = −
∑r−1

i=1 τi.

Coding for Explanatory Variables

We will define r− 1 variables Xk, k = 1, 2, . . . , r− 1.

Values of these variables will be denoted Xi,j,k, where

the i, j subscript refers to the case Yi,j (i = factor level,

j = # of observation at that level)

Xi,j,k =



















1 if Yi,j is from level k

−1 if Yi,j is from level r

0 if Yi,j is from any other level

Recall that our notation for Yi,j means that Yi,j is from

level i. Thus the X variables are

Xi,j,k =



















1 i = k

−1 i = r

0 i 6= k, r

i = 1, . . . , r, j = 1, . . . , ni, k = 1, . . . , r − 1

The k subscript refers to the column of the design matrix

(not including the column of 1’s), and the i, j subscripts

indicate the rows (same order as the Yi,j).

The regression model

Yi,j = β0 + β1Xi,j,1 + . . .+ βr−1Xi,j,r−1 + ǫi,j really

becomes

Yi,j = β0 + βi + ǫi,j, i = 1, . . . , r − 1

Yr,j = β0 − β1 − . . .− βr−1 + ǫr,j
when the X ’s are plugged in as 0, 1, or -1. Comparing

this to the factor effects model Yi,j = µ+ τi + ǫi,j we

can make the following equivalencies:

β0 ≡ µ =
r

∑

i=1

µi/r;

βi ≡ τi, i = 1, . . . , r − 1

τr ≡ −(β1 + . . . + βr−1) = −
r−1
∑

i=1

τi so that
∑r

i=1 τi = 0.

Thus, defining the indicator variables as we have done

also specifies the zero-sum constraint, which suggests

that µ is the mean of the group means, NOT the grand

mean (µ =
∑r

i=1 niµi/nT ).

KNNL Example

• KNNL page 706 (nknw698.sas )

• This is the “cereal box” example that we have

previously been working with. It is a bit messy

because ni = 5, 5, 4, 5. You have an easier example

in the homework (ni is constant).

• The grand mean is not the same as the mean of the

group means in this case since the n’s are different.

Look at the means

proc means data=cereal printalltypes;

class pkgdes;

var cases;

output out=cerealmeans mean=mclass;

The MEANS Procedure

Analysis Variable : cases

N

Obs N Mean Std Dev Minimum Maximum

------------------------------------------------------

19 19 18.6315789 6.4395525 10.0000000 33.0000000

------------------------------------------------------

Analysis Variable : cases

N

pkgdes Obs N Mean Std Dev Minimum Maximum

-------------------------------------------------------------

1 5 5 14.6000000 2.3021729 11.0000000 17.0000000

2 5 5 13.4000000 3.6469165 10.0000000 19.0000000

3 4 4 19.5000000 2.6457513 17.0000000 23.0000000

4 5 5 27.2000000 3.9623226 22.0000000 33.0000000

-------------------------------------------------------------

proc print data=cerealmeans;

Obs pkgdes _TYPE_ _FREQ_ mclass

1 . 0 19 18.6316

2 1 1 5 14.6000

3 2 1 5 13.4000

4 3 1 4 19.5000

5 4 1 5 27.2000

Note: Type = 0 indicates the grand mean, Type =

1 indicates the means are for the levels of a predictor

variable. Type = 2would indicate that we had two

predictor variables and for each a level was specified.

Explanatory Variables

Set X1 to be 1 for design 1, -1 for design 4, and 0

otherwise; X2 is 1 for design 2, -1 for design 4, and 0

otherwise; X3 is 1 for design 3, -1 for design 4, and 0

otherwise..

data cereal; set cereal;

x1=(pkgdes eq 1)-(pkgdes eq 4);

x2=(pkgdes eq 2)-(pkgdes eq 4);

x3=(pkgdes eq 3)-(pkgdes eq 4);

proc print data=cereal; run;

New Variables

Obs cases pkgdes store x1 x2 x3

1 11 1 1 1 0 0

2 17 1 2 1 0 0

3 16 1 3 1 0 0

4 14 1 4 1 0 0

5 15 1 5 1 0 0

6 12 2 1 0 1 0

7 10 2 2 0 1 0

8 15 2 3 0 1 0

9 19 2 4 0 1 0

10 11 2 5 0 1 0

11 23 3 1 0 0 1

12 20 3 2 0 0 1

13 18 3 3 0 0 1

14 17 3 4 0 0 1

15 27 4 1 -1 -1 -1

16 33 4 2 -1 -1 -1

17 22 4 3 -1 -1 -1

18 26 4 4 -1 -1 -1

19 28 4 5 -1 -1 -1

Interpret X ’s in terms of parameters

Note the µ is implicit just like the intercept

pkgdes Int x1 x2 x3

1 1 1 0 0 µ+ τ1

2 1 0 1 0 µ+ τ2

3 1 0 0 1 µ+ τ3

4 1 -1 -1 -1 µ− τ1 − τ2 − τ3

Run the regression

proc reg data=cereal;

model cases=x1 x2 x3;

The REG Procedure

Model: MODEL1

Dependent Variable: cases

Analysis of Variance

Sum of Mean

Source DF Squares Square Pr > F

Model 3 588.22105 196.07368 <.0001

Error 15 158.20000 10.54667

Corrected Total 18 746.42105

Root MSE 3.24756 R-Square 0.7881

Dependent Mean 18.63158 Adj R-Sq 0.7457

Parameter Estimates

Parameter Standard

Variable DF Estimate Error t Value Pr > |t|

Intercept 1 18.67500 0.74853 24.95 <.0001

x1 1 -4.07500 1.27081 -3.21 0.0059

x2 1 -5.27500 1.27081 -4.15 0.0009

x3 1 0.82500 1.37063 0.60 0.5562

Compare with proc glm

proc glm data=cereal;

class pkgdes;

model cases=pkgdes;

The GLM Procedure

Dependent Variable: cases

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 3 588.2210526 196.0736842 18.59 <.0001

Error 15 158.2000000 10.5466667

Corrected Total 18 746.4210526

R-Square Coeff Var Root MSE cases Mean

0.788055 17.43042 3.247563 18.63158

Source DF Type I SS Mean Square F Value Pr > F

pkgdes 3 588.2210526 196.0736842 18.59 <.0001

Interpret the Regression Coefficients

Var Est

Int 18.675 b0 = µ̂ (mean of the means)

x1 -4.075 b1 = τ̂1 = Ȳ1. − µ̂ (effect of level 1)

x2 -5.275 b2 = τ̂2 = Ȳ2. − µ̂ (effect of level 2)

x3 0.825 b3 = τ̂3 = Ȳ3. − µ̂ (effect of level 3)

b0 + b1 = 18.675− 4.075

= 14.6 (mean for level 1)

b0 + b2 = 18.675− 5.275

= 13.4 (mean for level 2)

b0 + b2 = 18.675 + 0.825

= 19.5 (mean for level 3)

b0 − b1 − b2 − b3 = 18.675 + 4.075 + 5.275− 0.825

= 27.2 (mean for level 4)

meansstatement output

Level of ------------cases------------

pkgdes N Mean Std Dev

1 5 14.6000000 2.30217289

2 5 13.4000000 3.64691651

3 4 19.5000000 2.64575131

4 5 27.2000000 3.96232255

Plot the means

proc means data=cereal;

var cases; by pkgdes;

output out=cerealmeans mean=avcases;

symbol1 v=circle i=join;

proc gplot data=cerealmeans;

plot avcases*pkgdes;

The means

Confidence Intervals

• Ȳi. ∼ N
(

µi,
σ2

ni

)

(since Yi,j ∼ N(µi, σ
2))

• CI for µi is Ȳi. ± tc
s√
ni

(remember s =
√
MSE,

s√
ni

is often called the standard error of the mean)

• tc is computed from the tnT−r(1− α
2 ) distribution.

CI’s using proc means

You can get CI’s from proc means , but it does not

use the above formula. Instead proc means uses

si√
ni

for the CI at level i (CI for µi). It uses the

within-group variation to estimate the standard error for

each level, and does not assume all levels have a

common variance. The df for tc is ni − 1 for level i.

Thus the CI’s using proc meanswill have different

widths depending on their si’s and ni’s.

proc means data=cereal

mean std stderr clm

maxdec=2;

class pkgdes;

var cases;

The MEANS Procedure

Analysis Variable : cases

N Lower 95% Upper 95%

pkgdes Obs Mean Std D Std Err CL Mean CL Mean

-----------------------------------------------------

1 5 14.60 2.30 1.03 11.74 17.46

2 5 13.40 3.65 1.63 8.87 17.93

3 4 19.50 2.65 1.32 15.29 23.71

4 5 27.20 3.96 1.77 22.28 32.12

-----------------------------------------------------

CI’s using proc glm

These use the pooled standard error formula (s not si)
and the df is nT − r as given in the formula above. This

is the way we will generally prefer since we have more

degrees of freedom due to the constant variance

assumption (and hence smaller MSE and SE’s).

proc glm data=cereal;

class pkgdes;

model cases=pkgdes;

means pkgdes/t clm;

The GLM Procedure

t Confidence Intervals for cases

Alpha 0.05

Error Degrees of Freedom 15

Error Mean Square 10.54667

Critical Value of t 2.13145

95% Confidence

pkgdes N Mean Limits

4 5 27.200 24.104 30.296

3 4 19.500 16.039 22.961

1 5 14.600 11.504 17.696

2 5 13.400 10.304 16.496

These CI’s are often narrower than the ones from proc

means because more degrees of freedom (common

variance). Notice that these CI’s are all the same width

except for design 3 (ni = 4). They are sorted by

descending mean. Here the glmCI is narrower for

designs 2, 3, and 4 but slightly wider for design 1.

(Design 1 had the smallest si = 1.03.)

Multiplicity Problem

• We have constructed 4 (in general, r) 95%

confidence intervals. So the overall (family)

confidence level (confidence that every interval

contains its mean) is less than 95%.

• Many different kinds of adjustments have been

proposed.

• We have previously discussed the Bonferroni

correction (i.e., use α/r).

bonoption in SAS

proc glm data=cereal;

class pkgdes;

model cases=pkgdes;

means pkgdes/bon clm;

The GLM Procedure

Bonferroni t Confidence Intervals for cases

Alpha 0.05

Error Degrees of Freedom 15

Error Mean Square 10.54667

Critical Value of t 2.83663

Simultaneous 95%

pkgdes N Mean Confidence Limits

4 5 27.200 23.080 31.320

3 4 19.500 14.894 24.106

1 5 14.600 10.480 18.720

2 5 13.400 9.280 17.520

Hypothesis Tests on Individual Means

Not common, but can be done.

Use proc means options t and probt for a test of

the null hypothesis H0 : µi = 0

To test H0 : µi = c, where c is an arbitrary constant,

first use a data step to subtract c from all observations

and then run proc means options t and probt .

proc means data=cereal mean std stderr clm maxdec=2;

class pkgdes;

var cases;

The MEANS Procedure

Analysis Variable : cases

N

pkgdes Obs Mean Std Dev Std Error t Value Pr > |t|

-----------------------------------------------------------

1 5 14.60 2.3021729 1.0295630 14.18 0.0001

2 5 13.40 3.6469165 1.6309506 8.22 0.0012

3 4 19.50 2.6457513 1.3228757 14.74 0.0007

4 5 27.20 3.9623226 1.7720045 15.35 0.0001

-----------------------------------------------------------

Can also use GLM’s mean statement with clm option

and see whether it contains 0 (or the hypothesized

value). This has the advantage of more df than the

proc meansway.

Differences in means

Ȳi. − Ȳk. ∼ N

(

µi − µk,
σ2

ni
+

σ2

nk

)

We can test for equality of means by testing whether this

difference is 0 (or looking to see whether 0 is in the CI).

CI for µi − µk is Ȳi. − Ȳk. ± tcs{Ȳi. − Ȳk.}, where

s{Ȳi. − Ȳk.} = s
√

1
ni
+ 1

nk
.

Topic VI

Page 75



Statistics 512: Applied Regression Analysis

Professor Min Zhang

Purdue University

Spring 2014

Multiple Comparisons: Determining the critical value

We deal with the multiplicity problem by adjusting tc.

Many different choices are available. These roughly fall

into two categories:

• Change α level.

• Use a different distribution.

We will consider 4 slightly different testing procedures.

LSD

• Least Significant Difference (LSD) - this is the “least conservative” procedure

we have.

• Simply ignores multiplicity issue and controls the test-alpha. If we have a lot

of tests, it becomes very likely that we will make Type I errors (reject when we

should not).

• Has better power than the rest of the tests.

• Uses tc = tnT−r(1− α
2 ).

• Called t or LSD in SAS.

• This procedure is really too liberal and is not one that we often use.

Bonferroni

• More conservative than Tukey, but better if we only want to do comparisons

for a small number of pairs of treatment means.

• Use the error budgeting idea to get family confidence level at least 1− α.

• Sacrifices a little more power than Tukey.

• There are





r

2



 = r(r−1)
2 comparisons among r means, so replace α

by
2α

r(r−1) and use tc = tnT−r(1− α
r(r−1)). For large r, Bonferroni is too

conservative.

• Called bon in SAS.

Tukey

• More conservative than LSD.

• Specifies exact family alpha-level for comparing all pairs of treatment means,

but has less power than LSD (wider CI’s).

• Based on the studentized range distribution (maximum minus minimum

divided by the standard deviation). See Table B.9.

• Uses tc =
qc√
2

.

• Details are in KNNL Section 17.5.

• Called tukey in SAS.

Scheffé

• Most conservative of the tests (sometimes).

• Controls family alpha level for testing ALL linear combinations of means (we’ll

talk about these later) but has less power (and so get CI’s that are too wide).

For testing pairs of treatment means it is (a bit) too conservative.

• Based on the F distribution

• tc =
√

(r − 1)Fr−1,nT−r(1− α)

• Protects against data snooping

• Called scheffe in SAS

Multiple Comparisons Summary

• LSD is too liberal (get Type I errors / CI’s too narrow)

• Scheffe is conservative (no power for certain

comparisons/ CI’s wide)

• Bonferroni is OK for small r (but conservative for

large r)

• Tukey (HSD) is recommended for general use

Examples

proc glm data=a1;

class pkgdes;

model cases=pkgdes;

means pkgdes/lsd tukey bon scheffe;

means pkgdes/lines tukey;

LSD
The GLM Procedure

t Tests (LSD) for cases

NOTE: This test controls the Type I comparisonwise error rate, not the

experimentwise error rate.

Alpha 0.05

Error Degrees of Freedom 15

Error Mean Square 10.54667

Critical Value of t 2.13145

Comparisons significant at the 0.05 level are indicated by ***.

Difference

pkgdes Between 95% Confidence

Comparison Means Limits

4 - 3 7.700 3.057 12.343 ***

4 - 1 12.600 8.222 16.978 ***

4 - 2 13.800 9.422 18.178 ***

3 - 4 -7.700 -12.343 -3.057 ***

3 - 1 4.900 0.257 9.543 ***

3 - 2 6.100 1.457 10.743 ***

1 - 4 -12.600 -16.978 -8.222 ***

1 - 3 -4.900 -9.543 -0.257 ***

1 - 2 1.200 -3.178 5.578

2 - 4 -13.800 -18.178 -9.422 ***

2 - 3 -6.100 -10.743 -1.457 ***

2 - 1 -1.200 -5.578 3.178

Tukey
The GLM Procedure

Tukey’s Studentized Range (HSD) Test for cases

NOTE: This test controls the Type I experimentwise error rate.

Alpha 0.05

Error Degrees of Freedom 15

Error Mean Square 10.54667

Critical Value of Studentized Range 4.07597

Comparisons significant at the 0.05 level are indicated by ***.

Difference

pkgdes Between Simultaneous 95%

Comparison Means Confidence Limits

4 - 3 7.700 1.421 13.979 ***

4 - 1 12.600 6.680 18.520 ***

4 - 2 13.800 7.880 19.720 ***

3 - 4 -7.700 -13.979 -1.421 ***

3 - 1 4.900 -1.379 11.179

3 - 2 6.100 -0.179 12.379

1 - 4 -12.600 -18.520 -6.680 ***

1 - 3 -4.900 -11.179 1.379

1 - 2 1.200 -4.720 7.120

2 - 4 -13.800 -19.720 -7.880 ***

2 - 3 -6.100 -12.379 0.179

2 - 1 -1.200 -7.120 4.720

Note
4.07√

2
= 2.88 is the tc value used to make the CI.

Output (lines option)

. Mean N pkgdes

Tukey Grouping

A 27.200 5 4

B 19.500 4 3

B

B 14.600 5 1

B

B 13.400 5 2

Run nknw711.sas for yourself to see and compare

the intervals using the lsd , bon and

scheffe options. They are wider than the

tukey intervals. However, all three corrected methods

(bon , tukey , scheffe ) ultimately give the same

conclusion for this example, namely, that design 4 has a

significantly higher mean than the other three, but

designs 1, 2, and 3 are not significantly different from

one another.

Some other options in proc glm

• alpha=0.xx either in the procedure statement or after ‘/ ’ in

the modelor means statement will change your alpha-level for

the respective statement(s).

• /DUNNETT(’Control’)will perform tests that compare

treatments to a control (where the ‘control’ in parentheses is the

name of the level which is the control). This has more power

than Tukey with the same family alpha in the case where you are

making only those comparisons.

• /LINESwill cause the tests to take a more convenient output

(see last example above).

Linear Combinations of Means

• Often we wish to examine (test hypotheses about, make CI’s for)

particular linear combinations of the group means.

• These combinations should come from research questions, not

from an examination of the data.

• A linear combination of means is any quantity of the form

L =
∑

i ciµi for any constants ci. We estimate L with

L̂ =
∑

i ciȲi. ∼ N(L, Var(L̂)).

• Its variance is Var(L̂) =
∑

i c
2

i Var(Ȳi), which can be estimated

by s2
∑

i

c2i
ni

.

Contrasts

• A contrast is a special case of a linear combination with
∑

i ci = 0. These

turn out to be particularly useful because the interesting hypothesis tests are

of the form H0 : L = 0.

• Example 1: µ1 − µ2 (c1 = 1, c2 = −1)

• Used to test whether levels 1 and 2 have equal means.

• Example 2: µ1 − 1
2 (µ2 + µ3) (1,−0.5,−0.5)

• Used to test whether level 1 has the same mean as the combination of levels

2/3.

• Example 3: (µ1 + µ2)/2− (µ3 + µ4)/2 (0.5, 0.5, −0.5, −0.5)

• Used to test whether the first two levels have the same mean as the last two

(think 1, 2 = men; 3, 4 = women and 1, 3 = diet A; 2, 4 = diet B - this

would then test for gender differences)

contrastand estimateoptions in SAS

For Example 3 above

proc glm data=a1;

class pkgdes;

model cases=pkgdes;

contrast ’1&2 v 3&4’ pkgdes .5 .5 -.5 -.5;

estimate ’1&2 v 3&4’ pkgdes .5 .5 -.5 -.5;

Contrast DF Contrast SS Mean Square F Value Pr > F

1&2 v 3&4 1 411.4000000 411.4000000 39.01 <.0001

Standard

Parameter Estimate Error t Value Pr > |t|

1&2 v 3&4 -9.35000000 1.49705266 -6.25 <.0001

The contrast statement performs the F -test. The estimate statement performs a

t-test and gives the parameter estimate.

Multiple Contrasts

• We can simultaneously test a collection of contrasts (1 df each

contrast)

• Example 1, H0 : µ1 = (µ2 + µ3 + µ4)/3

• The F statistic for this test will have an F1,nT−r distribution

• Example 2, H0 : µ2 = µ3 = µ4.

• The F statistic for this test will have an F2,nT−r distribution

• We do this by setting up one contrast for each comparison and

doing them simultaneously.

proc glm data=a1;

class pkgdes;

model cases=pkgdes;

contrast ’1 v 2&3&4’ pkgdes 1 -.3333 -.3333 -.3333;

estimate ’1 v 2&3&4’ pkgdes 3 -1 -1 -1 /divisor=3;

contrast ’2 v 3 v 4’ pkgdes 0 1 -1 0, pkgdes 0 0 1 -1;

Contrast DF Contrast SS Mean Square F Value Pr >

1 v 2&3&4 1 108.4739502 108.4739502 10.29 0.0059

2 v 3 v 4 2 477.9285714 238.9642857 22.66 <.0001

Standard

Parameter Estimate Error t Value Pr > |t|

1 v 2&3&4 -5.43333333 1.69441348 -3.21 0.0059

Chapter 18 – Diagnostics: Overview

We will take the diagnostics and remedial measures that we learned

for regression and adapt them to the ANOVA setting. Many things

are essentially the same, while some things require modification.

Residuals

• Predicted values are cell means: Ŷi,j = Ȳi..

• Residuals are the differences between the observed values and

the cell means ei,j = Yi,j − Ȳi..

Basic plots

• Plot the data vs the factor levels (the values of the explanatory

variables)

• Plot the residuals vs the factor levels

• Construct a normal quantile plot of the residuals

Notice that we are no longer checking for linearity since this is not

an assumption in ANOVA.

KNNL Example

• KNNL page 734 (nknw712.sas )

• Compare 4 brands of rust inhibitor (X has r = 4

levels)

• Response variable is a measure of the effectiveness

of the inhibitor

• There are 40 observations, 10 units per brand

(n = 10 constant across levels)

data rust;

infile ’H:\System\Desktop\CH17TA02.DAT’;

input eff brand;

Recode the factor: just to show they can be letters

instead of numbers if you want.

data rust; set rust;

if brand eq 1 then abrand=’A’;

if brand eq 2 then abrand=’B’;

if brand eq 3 then abrand=’C’;

if brand eq 4 then abrand=’D’;

proc print data=rust;

Store the residuals in dataset rustout .

proc glm data=rust;

class abrand;

model eff = abrand;

output out=rustout r=resid;

Residuals have the same syntax as in proc reg .

Plots

• Data versus the factor

• Residuals versus the factor (or predictor)

• Normal quantile plot of the residuals

Plot vs the factor

symbol1 v=circle i=none;

proc gplot data=rustout;

plot (eff resid)*abrand;

Data vs the factor

Residuals vs the factor

Normal quantile plot of the residuals

proc univariate data = rustout;

qqplot resid / normal (L=1 mu=est sigma=est);

Summary of plot diagnostics

Look for

• Outliers

• Variance that depends on level

• Non-normal errors

Plot residuals vs time and other variables

Homogeneity tests

Homogeneity of variance (homoscedasticity) is assumed in the

model. We can test for that.

H0 : σ2

1
= σ2

2
= . . . = σ2

r (constant variance)

HA : not all σ2

i are equal (non-constant variance)

• Several significance tests are available. Note that this was also

available in regression if each X has multiple Y observations

(this is usually true in ANOVA): see Section 3.6.

• Text discusses Hartley, modified Levene.

• SAS has several including Bartlett’s (essentially the likelihood

ratio test) and several versions of Levene.

ANOVA is robust with respect to moderate deviations from normality,

but ANOVA results can be sensitive to the homogeneity of variance

assumption. In other words we usually worry more about constant

variance than we do about normality. However, there is a

complication: some homogeneity tests are sensitive to the normality

assumption. If the normality assumption is not met, we may not be

able to depend on the homogeneity tests.

Levene’s Test

• Do ANOVA on the squared residuals.

• Modified Levene’s test uses absolute values of the

residuals. Modified Levene’s test is recommended

because it is less sensitive to the normality

assumption.

KNNL Example

KNNL page 783 nknw768.sas

Compare the strengths of 5 types of solder flux (X has

r = 5 levels)

Response variable is the pull strength, force in pounds

required to break the joint

There are 8 solder joints per flux (n = 8)

data solder;

infile ’H:\System\Desktop\CH18TA02.DAT’;

input strength type;

Modified Levene’s Test

proc glm data=wsolder;

class type;

model strength=type;

means type/hovtest=levene(type=abs);

Dependent Variable: strength

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 4 353.6120850 88.4030213 41.93 <.0001

Error 35 73.7988250 2.1085379

Corrected Total 39 427.4109100

R-Square Coeff Var Root MSE strength Mean

0.827335 10.22124 1.452081 14.20650

Topic VI

Page 107



Statistics 512: Applied Regression Analysis

Professor Min Zhang

Purdue University

Spring 2014

.Levene’s Test for Homogeneity of strength Variance

ANOVA of Absolute Deviations from Group Means

Sum of Mean

Source DF Squares Square F Value Pr > F

type 4 8.6920 2.1730 3.07 0.0288

Error 35 24.7912 0.7083

Rejecting H0 means there is evidence that variances are

not homogeneous.

Means and SD’s

The GLM Procedure

Level of -----------strength----------

type N Mean Std Dev

1 8 15.4200000 1.23713956

2 8 18.5275000 1.25297076

3 8 15.0037500 2.48664397

4 8 9.7412500 0.81660337

5 8 12.3400000 0.76941536

The standard deviations do appear quite different.

Remedies

• Delete outliers – Is their removal important?

• Use weights (weighted regression)

• Transformations

• Nonparametric procedures

Weighted Least Squares

• We used this with regression.

• Obtain model for how the sd depends on the explanatory

variable (plotted absolute value of residual vs x)

• Then used weights inversely proportional to the estimated

variance

• Here we can compute the variance for each level because we

have multiple observations (replicates).

• Use these as weights in proc glm

• We will illustrate with the soldering example from KNNL.

Topic VI

Page 111



Statistics 512: Applied Regression Analysis

Professor Min Zhang

Purdue University

Spring 2014

Obtain the variances and weights

proc means data=solder;

var strength;

by type;

output out=weights var=s2;

data weights;

set weights;

wt=1/s2;

NOTE Data set weights has 5 “observations”, one for

each level.

Merge and then use the weights in proc glm

data wsolder;

merge solder weights;

by type;

proc glm data=wsolder;

class type;

model strength=type;

weight wt;

The GLM Procedure

Dependent Variable: strength

Weight: wt

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 4 324.2130988 81.0532747 81.05 <.0001

Error 35 35.0000000 1.0000000

Corrected Total 39 359.2130988

R-Square Coeff Var Root MSE strength Mean

0.902565 7.766410 1.00000 12.87596

Note the increase in the size of the F -statistic as well as R2. Also

notice that the MSE is now 1.

Transformation Guides

Transformations can also be used to solve constant

variance problems, as well as normality.

• When σ2
i is proportional to µi, use

√
Y .

• When σi is proportional to µi, use log(Y ).

• When σi is proportional to µ2
i , use 1/Y .

• When Y is a proportion, use 2 arcsin(
√
Y ); this is

2*arsin(sqrt(y)) in a SAS data step.

• Can also use Box-Cox procedure.

Nonparametric approach

• Based on ranks

• See KNNL Section 18.7, page 795

• See the SAS procedure npar1way

Section 17.9: Quantitative Factors

• Suppose the factor X is a quantitative variable (has a numeric order to its

values).

• We can still do ANOVA, but regression is a possible alternative analytical

approach.

• Here, we will compare models (e.g., is linear model appropriate or do we

need quadratic, etc.)

• We can look at extra SS and general linear tests.

• We use the factor first as a continuous explanatory variable (regression) then

as a categorical explanatory variable (ANOVA)

• We do all of this in one run with proc glm

• This is the same material that we skipped when we studied regression: F

Test for Lack of Fit , KNNL Section 3.7, page 119.

KNNL Example

• KNNL page 762 (nknw742.sas )

• Y is the number of acceptable units produced from

raw material

• X is the number of hours of training

• There are 4 levels for X : 6 hrs, 8 hrs, 10 hrs and 12

hrs.

• i = 1 to 4 levels (r = 4)

• j = 1 to 7 employees at each training level (n = 7)

data training;

infile ’H:\System\Desktop\CH17TA06.DAT’;

input product trainhrs;

Replace trainhrs by actual hours; also quadratic.

data training; set training;

hrs=2*trainhrs+4;

hrs2=hrs*hrs;

Obs product trainhrs hrs hrs2

1 40 1 6 36

...

8 53 2 8 64

...

15 53 3 10 100

...

22 63 4 12 144

...

PROC GLMwith both categorical (“class ”) and

quantitative factors: if a variable is not listed on the

class statement, it is assumed to be quantitative, i.e.

a regression variable.

proc glm data=training;

class trainhrs;

model product=hrs trainhrs / solution;

Note the multicollinearity in this problem:

hrs = 12− 6X1 − 4X2 − 2X3 − 0X4. Therefore,

we will only get 3 (not 4) model df .

The GLM Procedure

Dependent Variable: product

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 3 1808.678571 602.892857 141.46 <.0001

Error 24 102.285714 4.261905

Corrected Total 27 1910.964286

R-Square Coeff Var Root MSE product Mean

0.946474 3.972802 2.064438 51.96429

Source DF Type I SS Mean Square F Value Pr > F

hrs 1 1764.350000 1764.350000 413.98 <.0001

trainhrs 2 44.328571 22.164286 5.20 0.0133

The Type I test for trainshrs looks at the lack of fit.

It asks, with hrs in the model (as a regression variable),

does trainhrs have anything to add (as an ANOVA)

variable? The null hypothesis for that test is that a

straight line model with hrs is sufficient. Although

hrs and trainhrs contain the same information,

hrs is forced to fit a straight line, while trainhrs can

fit any way it wants. Here it appears there is a significant

deviation of the means from the fitted line because

trainhrs is significant; the model fits better when

non-linearity is permitted.

Interpretation

The analysis indicates that there is statistically

significant lack of fit for the linear regression model

(F = 5.20; df = 2, 24; p = 0.0133)

Looking at the plot suggests there is some curvature to

the relationship. Let’s try a quadratic term in the model.

Quadratic Model

proc glm data=training;

class trainhrs;

model product=hrs hrs2 trainhrs;

The GLM Procedure

Dependent Variable: product

Sum of

Source DF Squares Mean Square F Value Pr > F

Model 3 1808.678571 602.892857 141.46 <.0001

Error 24 102.285714 4.261905

Corrected Total 27 1910.964286

R-Square Coeff Var Root MSE product Mean

0.946474 3.972802 2.064438 51.96429

Source DF Type I SS Mean Square F Value Pr > F

hrs 1 1764.350000 1764.350000 413.98 <.0001

hrs2 1 43.750000 43.750000 10.27 0.0038

trainhrs 1 0.578571 0.578571 0.14 0.7158

When we include a quadratic term for hrs , the remaining

trainhrs is not significant. This indicates that the quadratic

model is sufficient and allowing the means to vary in an arbitrary

way is not additionally helpful (does not fit any better). Note that the

lack of fit test now only has 1 df since the model df has not

changed.

